
Using Internal Tables: Unit Objectives

l

Define internal tables

l

Perform operations on internal tables

l

Identify table kinds and use them appropriately in 

different situations

At the conclusion of this unit, you will be able to:

· Internal tables are data objects that allow you to store datasets with a fixed structure in memory. The data is stored line by line in memory. Each line has the same structure.

· You can also refer to each component in a line as a column in the internal table. You refer to each line in the internal table as a table line or table entry.

· Internal tables are dynamic data objects - that is, they can hold any number of lines of a fixed type. The number of lines in an internal table is limited only by the capacity of the specific system with which you are working. 

· You can define the line type of an internal table to include elementary, structured, or even table types, and can be as complex as you want to make it. 

· Each internal table has the following attributes:

· Line type: 
You use the line type to specify which columns are to be included in your internal table. You must assign a name and type to each column. You generally define a line type using a structure type. 

· Key: 
Internal tables have a key, just like database tables. The key is defined by: Specifying the columns that should belong to the key; the sequence; and whether or not the key should be unique or non-unique.
 Depending on how you define your internal table, the runtime system may create an additional unique key (the index) for some internal tables (this will be discussed in more detail later). For simplicity's sake you can think of the index as a line number.  

· Table kind: 
There are three table kinds in ABAP. Each kind has different attributes, which specify whether the key is unique, how the index is managed, and how the table data is addressed internally. The table kind specifies the possible accesses. The purpose of this is to provide particular support for special access types from the runtime system (will be discussed later).

The table kind is linked to the access type as follows:

· You can only perform an index access (the fastest type of access) on STANDARD and SORTED tables.

· The runtime system implements key accesses of SORTED and HASHED tables in a way that optimizes runtime performance. Key access of STANDARD tables is by table scan - that is, comparing the field contents with the search key line by line, in a loop.

· The runtime system implements key accesses of SORTED tables, for fully or partly qualified keys (left-aligned, no gaps, all fields filled with an "=") in a special way to optimize runtime.
The same applies to partial sequential loops, where the loop condition is specified using the key fields. With the runtime-optimizing access, the runtime system keeps SORTED tables sorted by key fields, 
In all other cases, the system performs a table scan.

· The runtime optimizes key accesses of HASHED tables for fully qualified keys only. The table entries are addressed by means of a special hash algorithm. This minimizes access time for reading single records with fully qualified keys. The access time is thus independent of the number of lines in the table.
In all other cases, the system performs a table scan. This means that you should only use HASHED tables if you want to access table entries using a fully-qualified key.

· There are several ways to define an internal table in ABAP:

· Use the TYPES statement to define a table type. Then use this to assign a type to one or more data objects or interface variables.

· Alternatively, declare the table attributes directly when you define the data object (that is, use a bound type).

· The third option is to define an internal table using a global table type (defined in the ABAP Dictionary). 

· The tables you define using the first or second option are visible only in the program in which you define them. Whenever you need table type visible from all programs, create a global table type (that is, use the third option).

· To define a table type (explicitly or implicitly), you must give the type (or data object) a name, as well as specifying a table kind, line type, and key. 

· If you are defining a local table type, enter the kind after TYPE and the line type after OF. You must list the key fields after WITH. 

· You create and edit global types in the ABAP Dictionary. If you are defining a global table type, specify the same information on the maintenance screens.

· For table types defined in a program, you can enter the number of lines that the runtime system should reserve when it initializes the data object, after the INITIAL SIZE addition. This makes sense if you know exactly how many lines you will want in your table when you create it. However, if your table needs more lines at runtime, it will not be limited in size by this addition, since the runtime system frees the necessary memory dynamically.
(Internal tables are dynamic data objects). 

· To specify the line type of an internal table, you can use all the local and global data types or data objects.

· Internal tables are most frequently used to display contents of the database tables. Normally, non-nested structured data types are used for this.

· If you use a line type, you must use a statement in the form: TYPE table_kind OF line_type, where table_kind is the kind of table (to be discussed later) and line_type is the name of the structure type you are using.

· If you use structure (data object), you must use a statement in the form: LIKE table_kind OF line_type, where structure is the name of the structure object you are using.

· There are three parameters that you use to specify a table kind for your internal table: STANDARD TABLE, SORTED TABLE and HASHED TABLE
· You specify the table key with the WITH key_def addition. key_def includes the names of all the key fields in order and specifies whether the key is to be UNIQUE or NON-UNIQUE. 

· The combination of the table kind and the key definition is very significant, because of the special support that certain table kinds receive with specific types of read access. You can use any of the following combinations:

· For STANDARD tables: 
Either create a user-defined key by naming the key fields after NON-UNIQUE KEY, or specify the standard key using the WITH DEFAULT KEY addition. The standard key consists of all the fields with character-type data types  (c, n, d, t, x, string, xstring). 

· For SORTED tables: 
List the key fields after WITH UNIQUE KEY or NON-UNIQUE KEY as appropriate.

· For HASHED tables: 
List the key fields after WITH UNIQUE KEY.

· Alternatively, use the pseudo-component table_line, if you are specifying a table without a structured line type, or if the entire table line is being used as the key. This will be discussed in more detail later in the unit.

· The above example shows the definition of an internal table (tt_distance) using a local table type (tt_distance_ty), which itself uses a local line type (s_distance_ty).

· The internal table defined here is a STANDARD table with the line type s_distance_ty. It contains the columns carrid, connid, distance and distid. 

· The distance and distid fields are key fields. The key is not unique.

· The above example shows the definition of a SORTED table and a HASHED table. Both tables have the same line type as the STANDARD table on the previous slide. 

· Note that the contents of the table are in a different order. For SORTED tables, the sequence of the entries in the internal table is determined by the sequence of fields in the key definition. 

· A standard type exists for defining standard tables with bound types. (Table type STANDARD TABLE, key WITH DEFAULT KEY). When you create such a table, you can omit the STANDARD and WITH DEFAULT KEY additions, since the runtime system supplies them automatically.

· Note however, that the standard table type exists for data objects only. If you are defining a table type that you want to use to provide a type for data objects, you must specify all its attributes completely.

· Incomplete table types are known as generic. You can only use them to assign types to table-type interface parameters. 

· Use an unstructured line type if you need a single-column internal table. The slide shows an example of a single-column table with the line type field_name.

· To declare an explicit key, you must use the pseudo-component table_line.
· You can use internal tables with an unstructured line type in the following ABAP statement (among others):

· SET PF-STATUS .. EXCLUDING itab. (deactivate function codes dynamically)
· SPLIT .. INTO TABLE itab. (split a string dynamically)

· You can use a key access, for example, to ascertain whether a specific entry exists. 

You can perform three different types of operations on internal tables:

· Operations on the whole data object: 
Elementary operations available include MOVE (copy the content line by line), CLEAR (delete the content, but keep the memory allocated), and FREE (delete the content and release the memory). You can also use all the numerical comparison operators with internal tables in logical expressions. With these operators, the number of lines is compared first and then, if necessary, the line contents. (This is recursive for deep types). For more information, refer to the documentation on internal tables. 
You can use the SORT statement to sort the content of  STANDARD and HASHED tables. This statement allows you to sort the table by one or columns, specifying ascending or descending for each column. For more details, refer to the keyword documentation for the SORT statement.  
You can use the SELECT ... INTO TABLE statement to copy the contents of database tables directly into internal tables.

· Operations on lines: 
You can use the INSERT, READ, MODIFY, DELETE, and COLLECT statements to perform operations on one or more lines in the internal table. The syntax of the first four of these statements is dealt with in the slides that follow. For the syntax of the COLLECT statement, refer to the online documentation.

· Loop processing: 
You can use the LOOP AT ... ENDLOOP statement to process the lines of the internal table sequentially. The exact syntax for this is dealt with in the slides that follow.

· Almost all line operations and loop processing statements offer you several options for specifying the target set of records to be processed. You can specify these target records by declaring a key or index, by formulating a condition that applies to some of the columns, or by specifying an index interval. Your use of these options is limited only by the table kind you have chosen. 
Example: You cannot specify a line in a hashed table using an index.

· When accessing table rows, you have two possibilities for a number of line operations and for loop processing: You can either copy the data into a structure with the same type as the line type of the internal table, or you can set a pointer to a table line and access the data in that line directly. 

· In particular, when you insert table lines, you can pass the data to be inserted using another internal table. 

· Note that all statements either set a return value or trigger a runtime error. If a return code has been set, it is stored in the sy-subrc field after the statement has been executed. Runtime errors are only triggered if the data passed at runtime are crucial in determining whether or not the operation can be performed. 
Example: 
You try to insert a new line into a SORTED table using an index. If you insert the new line in exactly the right place in a sorted table, the system performs the operation. If not, the system returns a runtime error.

· You can create a pointer to a data object in ABAP using a field symbol. 

· First, declare a data object using the FIELD-SYMBOLS statement. This data object can contain a pointer to another data object at runtime. Where possible, you should give the field symbol the same type as the data object (TYPE i, in this example).

· Note that the angle brackets (<>) are part of the name of the field symbol: In this example, the name is <fs>. 
· To point a field symbol at a data object, you must assign it to the object data_object using the ASSIGN data_object TO <fs> statement.

· You can use the field symbol to access the content of the data object to which it points - either to read or to change this content.

· You can "redirect" a field symbol to a different data object at runtime using the ASSIGN statement.

· In the above example, a field symbol is assigned the line type of an internal table. This makes it possible to assign a table line to this field symbol. The syntax required for this is discussed later in this unit.

· After a field symbol has been assigned a line, it is also possible to access the individual column values of the assigned line.

· As well as being able to read the data contents, you can also change the contents of the individual components.

[image: image21.wmf]ã

SAP

AG 2002

Insert

wa

itab

* 

fill workarea

"

wa

"

INSERT wa

INTO TABLE itab

.

INSERT

INSERT

INTO TABLE

INTO TABLE

wa

itab

* 

fill workarea

"

wa

"

INSERT wa

INTO itab

INDEX n.

INSERT

INSERT

INTO

INTO

INDEX

INDEX

n

n

itab1

itab2

INSERT LINES OF itab1

[

FROM

n1 TO n2 ]

target

.

[n1]

[n2]

Only if itab1 is

an index table

INSERT LINES OF

INSERT LINES OF

INTO TABLE itab2

.

INTO itab2

[ INDEX n ].

n

target

n

INTO TABLE

INTO TABLE

INTO

INTO

Key from "

wa

" 

specifies 

the line


To insert lines into a table, use the INSERT statement:

· Using INSERT wa INTO TABLE itab, you insert the line whose data are stored in wa into the internal table. The record is inserted with reference to the table key. In STANDARD tables, the new line is appended to the end of the table.

· Using INSERT wa INTO itab INDEX n, you insert the data from wa into the internal table at the line with the index n. If n is an index value within the internal table, the lines with the index value greater than or equal to n are pushed downwards. If n-1 is greater than the number of lines in the internal table, the line is not inserted. 

· Always add new lines to a SORTED table using the key. If you try to insert a line into a SORTED table using the index, you will not succeed unless the line has exactly the same index number as it would have when correctly sorted. Otherwise, a runtime error occurs.  

· Use INSERT LINES OF itab1 INTO TABLE itab2 to insert the lines of the internal table itab1 into the internal table itab2. Also, if itab1 is an index table, you can limit the number of lines to be inserted using an index interval (FROM n1 TO n2).

· If your target table is an index table, you can use INSERT LINES OF itab1 INTO itab2 to insert the lines of the internal table itab1 into the internal table itab2. You can then specify the line index from which the lines are to be inserted using INDEX n. Also, if itab1 is an index table, you can limit the number of lines to be inserted using an index interval (FROM n1 TO n2).

· You can read single table lines from index tables using the READ TABLE itab INDEX n INTO wa statement. After INDEX, enter the index of the line you want to read. If the system was able to read the line, it sets the return code to sy-subrc = 0 and stores the data from the line in wa. The total number of lines is then in sy-tfill, the length of the lines in sy-tleng. If the system could not read the line, the return code sy-subrc is unequal to 0. In this case, the content of wa is not changed.

· Use the option TRANSPORTING addition to specify the columns for which you want to transport data:

· If you do not want to read any data, but simply want to ascertain whether or not line n exists (evaluating the return value sy-subrc), use the TRANSPORTING NO FIELDS addition.

· If you want to read some of the columns in the line only, specify them after TRANSPORTING. Separate each column name with a space.

· You can use the optional addition COMPARING f1 ... fm to ascertain whether or not the line to be read n has specific column contents: To do this, copy the value of  all columns that point to the records you want to read, into wa and list the columns after the COMPARING addition to the READ statement. If the system was able to read the line and if all the columns listed contain the values stored in wa, it sets the return code sy-subrc to zero. If the system was able to read the line, but if one or more of the columns does not contain the value stored for it in wa, it sets the return code sy-subrc to two. If it could not read the line, it sets the return code to greater than two.

· The COMPARING ALL FIELDS addition provides a shorter syntax for comparing all the columns.

· You can read individual lines from any kind of table using READ TABLE itab key INTO wa. Use either a table key, or a comparison for some of the columns, as the search criterion key.

· If you want to use a table key, you have two options:

· Copy the key field values of the entry you want to read into the work area wa1 and use READ TABLE itab FROM wa1 INTO wa. Provided sy-subrc = 0, the system stores the result of the READ statement in the work area wa. Note that the values in wa1 that you have not explicitly filled contain appropriately-typed initial values.  Thus, the READ TABLE itab FROM wa1 statement searches for a line that has initial values in the key fields that were not declared explicitly.   
Note: 
You can also use a single work area wa for both declaring the key fields and receiving the result: 
READ TABLE itab FROM wa INTO wa. 

· Evaluate the key fields explicitly using the call READ TABLE itab WITH TABLE KEY k1 = f1 ... kn = fn statement. In this case, you must fill all the key fields.

· You can specify a formulated search condition to be applied to any columns using READ TABLE itab WITH KEY .... 

· However, you can only use dynamically formulated read accesses using the key and the READ TABLE itab FROM wa statement.

· You can use either a work area or a field symbol to access the individual lines in a table you want to read.

· To use a field symbol, first define it using FIELD-SYMBOLS <fs>. Give it the same type as the line type of the internal table that you want to read.
Example: FIELD-SYMBOLS <fs> LIKE LINE OF itab.

· Set the pointer to the correct line using the ASSIGNING <fs> addition instead of INTO. You can use any search criterion you want.

· You can access the components of the structure directly using the component names in the line type:

· You can directly address elementary components in the line type using  <fs>-field_1.

· If the line type contains structured components (a structure or internal table),
<fs>-component_name points to the entire structured component. To access the sub-components of this structured component, you need another field symbol.  

· Note: 
If you are searching for a specific string in an internal table, you can also use the SEARCH statement. For more details, refer to the keyword documentation for the SEARCH statement. 

· The MODIFY TABLE itab FROM wa statement allows you to change the content of one line of the internal table itab. The runtime system specifies the line to be changed using the key values from the work area wa and changes the non-key fields using the other fields. If your table has a non-unique key, the system changes the first entry only (using a linear search algorithm in STANDARD tables, and a binary search algorithm in SORTED tables). If you want to change only some of the fields (that is, columns) in a line, you must specify these after TRANSPORTING.

· You can change the nth line in an index table using MODIFY itab FROM wa INDEX n. Note that all the fields of the structure wa will be copied. Since changes to key fields in SORTED and HASHED tables can cause non-catchable runtime errors, you must use the TRANSPORTING addition in this case.  

· If you want to make the same changes to several lines in a table, use the MODIFY itab FROM wa TRANSPORTING f1 f2 ... WHERE log_expr statement. Specify all the columns to be changed after TRANSPORTING. You must place the new values for these columns in the work area wa. Specify a condition for the line using the WHERE clause.

· Instead of changing an individual line in a table using MODIFY, you can also change it using field symbols. First use READ TABLE ... ASSIGNING <fs> to assign a field symbol to the line to be changed. Then change the components directly using the field symbol components <fs>-field_1 = ....

· When you assign the field symbol using READ TABLE ... ASSIGNING you can specify the line to be read using a key (WITH TABLE KEY), index (for index tables), or a condition (WITH KEY).

· Note that you cannot change key fields in SORTED or HASHED tables. - trying to do so causes a runtime error.

· Use the READ TABLE itab INDEX n INTO wa statement to delete single table lines from index tables. Specify the line you want to delete using the table key. There are two ways of doing this:

· Copy the key field values of the entry you want to delete into the work area wa and use DELETE TABLE itab FROM wa.

· Fill the key fields directly in the DELETE statement using the WITH TABLE KEY k1 = f1 ... kn = fn addition.

· You can delete the nth line from the index table itab using DELETE itab INDEX n
· If you want to delete several lines from an internal table, use the  DELETE itab WHERE log_expr statement. The condition that specifies the lines you want to delete is declared in log_expr, in the WHERE clause (where log_expr is any logical expression applied to the columns).

· Use the LOOP AT itab result ... ENDLOOP statement to perform loop processing on the lines of the internal table itab. The system then executes the statement block between LOOP AT itab ... and ENDLOOP for each loop pass. result stands for:

· INTO wa: 
The system copies the table line it has processed in each loop pass into the work area wa
· ASSIGNING <fs>: 
The relevant line is assigned to the field symbol <fs>.

· With both variants you can use the TRANSPORTING NO FIELDS addition. No data is copied. Use this addition if, for example, you simply want to determine the number or indexes of the lines processed by the loop.

· Specify the number of lines that the loop is to process using a WHERE clause (condition for any columns).

· In an index table, limit the number of lines processed in the loop by declaring an index interval.

· While the loop is being processed, the system field sy-tabix contains the index of the table row of the current loop pass. The total number of lines is in sy-tfill, the length of the lines in sy-tleng.

· Note that in index tables, you can perform all the line operations (which you previously performed on the line specified in the INDEX n addition) without this addition in the loop. The operation is then performed on the current line.

· Index accesses (such as APPEND, INSERT ... INDEX, LOOP AT ... FROM ... TO) are possible for standard and sorted tables. However, caution is advised when using INSERT or APPEND on sorted tables;  this would violate the sort sequence and cause a runtime error.

· You can use the SORT statement to sort standard and hashed tables. Sorted tables are sorted by the runtime system.

· You can use key accesses with any table type, but their effect differs. A key access with INSERT has the same effect on standard tables as an APPEND, that is the relevant line is appended to the end of the table. In a sorted table however, the record is inserted in accordance with the sort sequence. In the case of a hashed table, the line is appended, but the system also changes the hash index internally.

· As well as the internal tables we have discussed, you can define internal tables with a header. These consist of a pair of components - the internal table itself (the body) and the work area (header), with the same line type. The header and the body have the same name, which sometimes simplifies the syntax for table accesses.  

· There are several possible syntax variants used to define internal tables with header lines, some of which are shown in the slide. 

· Internal tables with header lines are the oldest form of internal table.

· However, note that

· Many statements have a different effect on internal tables with a header than they would on normal internal tables. You can then address the body separately using "[]" .
Example: 
The CLEAR itab statement initializes only the header line, whereas in an internal with no header, it initializes the content of the entire table. 
You can delete the body of an internal table with a header using CLEAR itab[] or REFRESH itab.

· You cannot use internal tables with a header line in an object-oriented environment - that is, within classes or interfaces. 

· Use the SELECT-OPTION statement to create an internal table with a header, for which the runtime system automatically creates an input dialog for value sets for a selection screen. The system automatically inserts the appropriate entries in the internal table, from the user input. For more details, refer to the keyword documentation for the SELECT-OPTIONS statement. 

· As of SAP R/3 Basis Release 4.6A onwards, you can use the RANGE OF addition to the TYPES and DATA statements to define a corresponding (internal) table without a header line.

· Whenever you want to read individual table lines by declaring a complete key, use the READ TABLE ... WITH TABLE KEY statement (fastest single record access by key). The runtime system supports this syntax variant especially for SORTED and HASHED tables. If the table is a STANDARD  table, the runtime system performs a table scan.
The same applies if you have copied the values from all key fields of the entry to be read into the work area wa and are then use READ TABLE itab FROM wa.

· The runtime system carries out the syntax variant READ TABLE ... WITH KEY (read an entry after applying any condition) using a table scan.
The only exception to this rule applies to SORTED tables, if you fill the first n key fields with "=" (no gaps), where n <= number of key fields.
With standard tables however, you can also sort correspondingly using SORT and then use the BINARY SEARCH addition.

Summary:

· Whenever possible, use READ TABLE ... WITH TABLE KEY or the variant with a correspondingly-filled work area.

· If you need to use READ TABLE ... WITH KEY, make your internal table a SORTED table.

[image: image35.wmf]ã

SAP

AG 2002

Loop Processing and Table Kinds

LOOP AT

itab

INTO

wa

WHERE 

log_

expr

.

...

ENDLOOP.

log_

log_

expr

expr

Table scan

STANDARD

First n key fields filled

First n key fields filled

with "=" without gaps

with "=" without gaps

Any logical expression

Any logical expression

for columns

for columns

log_

expr

Table kind

Binary search for 

starting point, 

then loop only 

through group 

level

Table scan

SORTED

Table scan

HASHED


· The runtime system generally processes loops with a WHERE clause by performing a table scan - that is, determining whether the condition in the WHERE clause is true for each line in the table.
· SORTED tables are the only exception to this rule. For these, the runtime system optimizes the runtime under the following condition: 
In the WHERE clause, the first n key fields are filled with a "=" (no gaps). (n is less than or equal to the number of all key fields). As a result, the loop is only performed on the lines that match the condition in the WHERE clause. Since the table is sorted, the first line can be specified to optimize performance at runtime (using a binary search).

· To recap: 
If you program a loop through a standard table to match a specific field criterion, the runtime system always executes a table scan.

· You can use the algorithm described here to program the runtime behavior of sorted tables: 
First you must sort the standard table by the desired criterion, so that you can subsequently ascertain the starting and end points in a binary search. 
(The line indexes are available in the system field sy-tabix.)
Finally, you can use these values to program an index-controlled loop.

· Since the SORT and READ TABLE statements require additional runtime, this procedure is only useful if the loop can be repeated several times according to the field criterion.

· Summary:Use SORTED  tables if you want to implement partial sequential loops on internal tables (where the first n key fields are filled with "=") or use the above algorithm. 

[image: image37.wmf]ã

SAP

AG 2002

Access Using Field Symbols

<fs1>

<fs2>

READ TABLE itab 

ASSIGNING <fs1>

WITH TABLE KEY ... .   

READ TABLE <fs1>

-

col3

ASSIGNING <fs2>

WITH TABLE KEY ... .

wa1

wa2

Copy

Copy

col1

col2

col3

col4

col1

col2

col3

col4

itab

READ TABLE itab 

INTO wa1

WITH TABLE KEY ... .   

READ TABLE wa1

-

col3

INTO wa2

WITH TABLE KEY ... .

ASSIGNING

ASSIGNING

ASSIGNING

ASSIGNING

INTO

INTO

INTO

INTO


· Instead of READ TABLE ... INTO, you can use the READ TABLE ... ASSIGNING variant. This offers better performance at runtime for pure read accesses with a line width greater than or equal to 1000 bytes. If you then change the read line using MODIFY, READ ... ASSIGNING already improves runtime with a line width of 100 bytes.

· The same applies to LOOP ... INTO in comparison with LOOP ... ASSIGNING. The LOOP ... ASSIGNING variant offers better performance at runtime for any loop of five loop passes or more.

· Both field symbol variants are much faster than work area variants, in particular when you use nested internal tables. This is because, if you use work areas instead, the whole inner internal table is copied (unless you prevent this by using a TRANSPORTING addition).

· Always assign a type to field symbols, if you know their static type (again, for performance reasons).

· Note: 
If you use READ TABLE ... ASSIGNING the field symbol points to the originally assigned table line, even after the internal table has been sorted.

· Note that when using field symbols, you cannot change key fields in SORTED or HASHED tables. Trying to do so causes a runtime error.

· The following restrictions apply to LOOP ... ASSIGNING <fs>:

· You cannot use the SUM statement in control level processing. 

· You cannot reassign field symbols within the loop. The statements ASSIGN do TO <fs> and UNASSIGN <fs> will cause runtime errors. 

Using Internal Tables: Unit Summary


 Using Internal Tables Exercise 1

	Unit:
Using Internal Tables

Topic:
Single-Column Tables of the Type string

	At the conclusion of these exercises, you will be able to:

· Define single-column tables of the data type string
· Split character strings with internal tables

· Process internal tables in loops

	Now, the character string will contain more than one data record.

Extend your program in such a way that the various data records in the rows of an internal table are split. Then separate the individual data records into single components and output them as before.

	Program:


ZBC401_##_SPLIT_ITAB
Template:

SAPBC401_DTOS_SPLIT_STRING
Model solution:

SAPBC401_TABS_SPLIT_ITAB
## is your two-digit group number


1-1
Copy your solution for the exercise in the chapter Data Types and Data Objects in Detail, ZBC401_##_SPLIT_STRING, or the corresponding model solution SAPBC401_DTOS_SPLIT_STRING and give it the new name ZBC401_##_SPLIT_ITAB.

1-2
Ensure that the function module BC401_GET_SEP_STRING creates a character string consisting of 30 data records. For this you must assign a suitably typed constant (suggested name c_number) to the parameter im_number.

1-3
Define a single-column internal table of the type Standard. The column component is to have the data type string (suggested name: it_sets).

1-4
Change the SPLIT statement so that always one partial character string that contains a data record is placed into a row of its internal table.

1-5
Ensure that your internal table containing the partial character strings is used for the subsequent splitting of the individual data record and output of the components in a loop.

Exercise 2 - Optional

	Unit:
Using Internal Tables

Topic:
Table Types

	At the conclusion of these exercises, you will be able to:

· Select the appropriate types of internal table

· Process data using internal tables

	Now the character string will contain duplicate data records.

Extend your program in such a way that the various data records are kept sorted in internal tables. Those data records that appear twice are to be sorted into a separate internal table.

	Program:


ZBC401_##_TABKIND
Template:

SAPBC401_TABS_SPLIT_ITAB
Model solution:

SAPBC401_TABS_TABKIND
## is your two-digit group number


2-1
Copy your solution to the last exercise ZBC401_##_SPLIT_ITAB or the corresponding model solution SAPBC401_TABS_SPLIT_ITAB and give it the new name ZBC401_##_TABKIND.

2-2
Ensure that the function module BC401_GET_SEP_STRING creates a character string containing duplicate data records. For this, assign the parameter im_unique the value space.
Test your program.

2-3
Define an internal table in such a way that it can contain flight data sorted by airline, flight number, and flight date (suggested name it_flights).
Insert the separated data records (contents of the structure wa_flight) into this internal table, instead of displaying them. (You can still use the output statement later.)
Make use of the return value sy-subrc for the INSERT statement to ascertain whether or not a data record appears twice. For this, you must have defined the key of your internal table accordingly.
Define an additional internal table for the data records that appear twice (suggested name it_doubles).
Insert the duplicate data records into this internal table.

2-4
Display the contents of both internal tables with the flight data.

Exercise 3 - Optional

	Unit:
Using Internal Tables

Topic:
Processing Data Using Internal Tables

	At the conclusion of these exercises, you will be able to:

· Select the appropriate types of internal table

· Process data using internal tables

	Extend your program so that it displays the flight data sorted by the flight date. The data is to be displayed in different colors, according to a key date selected by the user.

If desired, the data is to be displayed using the SAP Grid Control.

	
	Program:


ZBC401_##_PROCESS_DATA
Template:

SAPBC401_TABS_TABKIND
Model solution:

SAPBC401_TABS_PROCESS_DATA
## is your two-digit group number


3-1
Copy your solution to the last exercise ZBC401_##_TABKIND or the corresponding model solution SAPBC401_TABS_TABKIND and give it the new name ZBC401_##_PROCESS_DATA.

3-2
Define a selection screen parameter for entering the key date (suggested name pa_date).
Specify the default value of the key date as 30 days in the future.
The user must still be able to choose the key date. Error message 085 of the message class bc401 is to be displayed if the selected key date is in the past.

3-3
Make sure that the data is first displayed sorted by flight date.

3-4
Ensure that the individual flight dates of the internal table it_flights are displayed in color according to the following criteria: Use the FORMAT COLOR col_... statement and load the type group col. (You can reverse any color set using the FORMAT RESET statement.)

· Display flights that occurred in the past with the background color col_negative.

· Display flights that occurred between today’s data and the key date with the background color col_total. 

· Display flights that occurred in the past with the background color col_positive.

3-5
Define a selection screen check box for displaying the data with the standard tool SAP Grid Control (suggested name pa_alv, addition AS CHECKBOX).
Ensure that the data is only displayed in the usual ABAP list if the user does not check the SAP Grid Control box.

Otherwise, proceed as follows:


3-5-1
Call the function module BC401_ALV_LIST_OUTPUT. Terminate the program if the function module raises an exception. (In this case, there is a problem with the system configuration.)

	Detailed information on using the SAP Grid Control is not part of this course. For this reason, you use a function module that encapsulates all the necessary technical details, which you can treat as a “black box.” 

For more information on the SAP Grid Control, see:

The online documentation

The example programs under
Environment ( Examples ( Control Examples 
The training course BC412: ABAP Dialog Programming with EnjoySAP Controls


3-5-2
Find out about the types of the two internal tables that can/must be transferred to the function module as a parameter.
Define two corresponding internal tables and an appropriate work area in your program (suggested names it_col_flights, it_col_doubles, and wa_col_flight).

3-5-3
The line type of the internal table that you must/can transfer contains a column COLOR, in which you can set the color values col_negative, col_total, col_positive, and col_background for every single line.
Copy the contents of the tables it_flights and it_doubles to the internal tables it_col_flights and it_col_doubles respectively. Use two loops to do this, and to fill the COLOR column.
it_flights:

	Logical condition
	Color value

	fldate < sy-datum
	col_negative

	fldate BETWEEN sy-datum AND p_date
	col_total

	fldate > p_date
	col_positive



Always use col_background to display it_doubles.

0.2 Using Internal Tables Solution 1
	[image: image51.png]



	Unit:
Using Internal Tables
Topic:
Single-Column Tables of the Type string


REPORT  sapbc401_tabs_split_itab.
TYPES:
  BEGIN OF st_flight_c,
    mandt(3)       TYPE c,
    carrid(3)      TYPE c,
    connid(4)      TYPE n,
    fldate(8)      TYPE n,
    price(20)      TYPE c,
    currency(5)    TYPE c,
    planetype(10)  TYPE c,
    seatsmax(10)   TYPE n,
    seatsocc(10)   TYPE n,
    paymentsum(22) TYPE c,
    seatsmax_b(10) TYPE n,
    seatsocc_b(10) TYPE n,
    seatsmax_f(10) TYPE n,
    seatsocc_f(10) TYPE n,
  END OF st_flight_c,
  BEGIN OF st_flight,
    carrid(3)     TYPE c,
    connid(4)     TYPE n,
    fldate        TYPE d,
    price(9)      TYPE p DECIMALS 2,
    currency(5)   TYPE c,
    planetype(10) TYPE c,
    seatsmax      TYPE i,
    seatsocc      TYPE i,
  END OF st_flight.
CONSTANTS c_number TYPE i VALUE 30.
DATA:
  datastring  TYPE string,
  set_string  TYPE string,
  wa_flight_c TYPE st_flight_c,
  wa_flight   TYPE st_flight.
DATA:
  it_sets TYPE STANDARD TABLE OF string
          WITH NON-UNIQUE DEFAULT KEY
          INITIAL SIZE c_number.
START-OF-SELECTION.
  CALL FUNCTION 'BC401_GET_SEP_STRING'
    EXPORTING
      im_number           = c_number
*      IM_TABLE_NAME       = 'SFLIGHT'
*      IM_SEPARATOR        = '#'
*      IM_UNIQUE           = 'X'
    IMPORTING
      ex_string           = datastring
    EXCEPTIONS
      no_data             = 1
      OTHERS              = 2.
  IF sy-subrc <> 0.
    MESSAGE a038(bc401).
  ENDIF.
  SHIFT datastring BY 2 PLACES IN CHARACTER MODE.
  FIND '##' IN datastring.
  IF sy-subrc <> 0.
    MESSAGE a702(bc401).
  ENDIF.
  SPLIT datastring AT '##' INTO TABLE it_sets.
  LOOP AT it_sets INTO set_string.
    SPLIT set_string AT '#' INTO
          wa_flight_c-mandt
          wa_flight_c-carrid
          wa_flight_c-connid
          wa_flight_c-fldate
          wa_flight_c-price
          wa_flight_c-currency
          wa_flight_c-planetype
          wa_flight_c-seatsmax
          wa_flight_c-seatsocc
          wa_flight_c-paymentsum.
    MOVE-CORRESPONDING wa_flight_c TO wa_flight.
    WRITE: /
      wa_flight-carrid,
      wa_flight-connid,
      wa_flight-fldate DD/MM/YYYY,
      wa_flight-price CURRENCY wa_flight-currency,
      wa_flight-currency,
      wa_flight-planetype,
      wa_flight-seatsmax,
      wa_flight-seatsocc.
  ENDLOOP.
Solution 2 - Optional

	Unit:
Using Internal Tables
Topic:
Table Types


REPORT  sapbc401_tabs_tabkind.
TYPES:
  BEGIN OF st_flight_c,
    mandt(3)       TYPE c,
    carrid(3)      TYPE c,
    connid(4)      TYPE n,
    fldate(8)      TYPE n,
    price(20)      TYPE c,
    currency(5)    TYPE c,
    planetype(10)  TYPE c,
    seatsmax(10)   TYPE n,
    seatsocc(10)   TYPE n,
    paymentsum(22) TYPE c,
    seatsmax_b(10) TYPE n,
    seatsocc_b(10) TYPE n,
    seatsmax_f(10) TYPE n,
    seatsocc_f(10) TYPE n,
  END OF st_flight_c,
  BEGIN OF st_flight,
    carrid(3)     TYPE c,
    connid(4)     TYPE n,
    fldate        TYPE d,
    price(9)      TYPE p DECIMALS 2,
    currency(5)   TYPE c,
    planetype(10) TYPE c,
    seatsmax      TYPE i,
    seatsocc      TYPE i,
  END OF st_flight.
CONSTANTS c_number TYPE i VALUE 30.
DATA:
  datastring  TYPE string,
  set_string  TYPE string,
  wa_flight_c TYPE st_flight_c,
  wa_flight   TYPE st_flight.
DATA:
  it_sets TYPE STANDARD TABLE OF string
          WITH NON-UNIQUE DEFAULT KEY
          INITIAL SIZE c_number,
  it_flights TYPE SORTED TABLE OF st_flight
             WITH UNIQUE KEY carrid connid fldate
             INITIAL SIZE c_number,
  it_doubles TYPE SORTED TABLE OF st_flight
             WITH NON-UNIQUE KEY carrid connid fldate
             INITIAL SIZE c_number.
START-OF-SELECTION.
  CALL FUNCTION 'BC401_GET_SEP_STRING'
     EXPORTING
       im_number           = c_number
*      IM_TABLE_NAME       = 'SFLIGHT'
*      IM_SEPARATOR        = '#'
       IM_UNIQUE           = space
     IMPORTING
       ex_string           = datastring
     EXCEPTIONS
       no_data             = 1
       OTHERS              = 2.
  IF sy-subrc <> 0.
    MESSAGE a038(bc401).
  ENDIF.
  SHIFT datastring BY 2 PLACES IN CHARACTER MODE.
  FIND '##' IN datastring.
  IF sy-subrc <> 0.
    MESSAGE a702(bc401).
  ENDIF.
  SPLIT datastring AT '##' INTO TABLE it_sets.
  LOOP AT it_sets INTO set_string.
    SPLIT set_string AT '#' INTO
          wa_flight_c-mandt
          wa_flight_c-carrid
          wa_flight_c-connid
          wa_flight_c-fldate
          wa_flight_c-price
          wa_flight_c-currency
          wa_flight_c-planetype
          wa_flight_c-seatsmax
          wa_flight_c-seatsocc
          wa_flight_c-paymentsum.
    MOVE-CORRESPONDING wa_flight_c TO wa_flight.
    INSERT wa_flight INTO TABLE it_flights.
    IF sy-subrc <> 0.
      INSERT wa_flight INTO TABLE it_doubles.
    ENDIF.
  ENDLOOP.
* output:
*********

  LOOP AT it_flights INTO wa_flight.
    WRITE: /
      wa_flight-carrid,
      wa_flight-connid,
      wa_flight-fldate DD/MM/YYYY,
      wa_flight-price CURRENCY wa_flight-currency,
      wa_flight-currency,
      wa_flight-planetype,
      wa_flight-seatsmax,
      wa_flight-seatsocc.
  ENDLOOP.
  SKIP.
  WRITE: / 'duplicate data records:'(dob) COLOR COL_HEADING.
  LOOP AT it_doubles INTO wa_flight.
    WRITE: /
      wa_flight-carrid,
      wa_flight-connid,
      wa_flight-fldate DD/MM/YYYY,
      wa_flight-price CURRENCY wa_flight-currency,
      wa_flight-currency,
      wa_flight-planetype,
      wa_flight-seatsmax,
      wa_flight-seatsocc.
  ENDLOOP.
Solution 3 - Optional

	Unit:
Using Internal Tables
Topic:
Processing Data Using Internal Tables


REPORT sapbc401_tabs_process_data.
TYPE-POOLS col.
TYPES:
  BEGIN OF st_flight_c,
    mandt(3)       TYPE c,
    carrid(3)      TYPE c,
    connid(4)      TYPE n,
    fldate(8)      TYPE n,
    price(20)      TYPE c,
    currency(5)    TYPE c,
    planetype(10)  TYPE c,
    seatsmax(10)   TYPE n,
    seatsocc(10)   TYPE n,
    paymentsum(22) TYPE c,
    seatsmax_b(10) TYPE n,
    seatsocc_b(10) TYPE n,
    seatsmax_f(10) TYPE n,
    seatsocc_f(10) TYPE n,
  END OF st_flight_c,
  BEGIN OF st_flight,
    carrid(3)     TYPE c,
    connid(4)     TYPE n,
    fldate        TYPE d,
    price(9)      TYPE p DECIMALS 2,
    currency(5)   TYPE c,
    planetype(10) TYPE c,
    seatsmax      TYPE i,
    seatsocc      TYPE i,
  END OF st_flight.
CONSTANTS c_number TYPE i VALUE 30.
DATA:
  datastring  TYPE string,
  set_string  TYPE string,
  wa_flight_c TYPE st_flight_c,
  wa_flight   TYPE st_flight.
DATA:
  it_sets TYPE STANDARD TABLE OF string
          WITH NON-UNIQUE DEFAULT KEY
          INITIAL SIZE c_number,
  it_flights TYPE SORTED TABLE OF st_flight
             WITH UNIQUE KEY fldate carrid connid
             INITIAL SIZE c_number,
  it_doubles TYPE SORTED TABLE OF st_flight
             WITH NON-UNIQUE KEY fldate carrid connid
             INITIAL SIZE c_number,
  it_col_flights TYPE bc401_t_flights_color,
  it_col_doubles LIKE it_col_flights,
  wa_col_flight LIKE LINE OF it_col_flights.
PARAMETERS:
  pa_date LIKE sy-datum,
  pa_alv  AS CHECKBOX DEFAULT 'X'.
LOAD-OF-PROGRAM.
  pa_date = sy-datum + 30.
AT SELECTION-SCREEN.
  IF pa_date < sy-datum.
    MESSAGE e085(bc401).               " date in the past
  ENDIF.
START-OF-SELECTION.
  CALL FUNCTION 'BC401_GET_SEP_STRING'
     EXPORTING
       im_number           = c_number
*      IM_TABLE_NAME       = 'SFLIGHT'
*      IM_SEPARATOR        = '#'
       im_unique           = space
     IMPORTING
       ex_string           = datastring
     EXCEPTIONS
       no_data             = 1
       OTHERS              = 2.
  IF sy-subrc <> 0.
    MESSAGE a038(bc401).
  ENDIF.
  SHIFT datastring BY 2 PLACES IN CHARACTER MODE.
  FIND '##' IN datastring.
  IF sy-subrc <> 0.
    MESSAGE a702(bc401).
  ENDIF.
  SPLIT datastring AT '##' INTO TABLE it_sets.
  LOOP AT it_sets INTO set_string.
    SPLIT set_string AT '#' INTO
          wa_flight_c-mandt
          wa_flight_c-carrid
          wa_flight_c-connid
          wa_flight_c-fldate
          wa_flight_c-price
          wa_flight_c-currency
          wa_flight_c-planetype
          wa_flight_c-seatsmax
          wa_flight_c-seatsocc
          wa_flight_c-paymentsum.
    MOVE-CORRESPONDING wa_flight_c TO wa_flight.
    INSERT wa_flight INTO TABLE it_flights.
    IF sy-subrc <> 0.
      INSERT wa_flight INTO TABLE it_doubles.
    ENDIF.
  ENDLOOP.
* output:
*********

  IF pa_alv = 'X'.
    LOOP AT it_flights INTO wa_flight.
      MOVE-CORRESPONDING wa_flight TO wa_col_flight.
      IF wa_col_flight-fldate < sy-datum.
        wa_col_flight-color = col_negative.
      ELSEIF wa_col_flight-fldate < pa_date.
        wa_col_flight-color = col_total.
      ELSE.
        wa_col_flight-color = col_positive.
      ENDIF.
      INSERT wa_col_flight INTO TABLE it_col_flights.
    ENDLOOP.
    LOOP AT it_doubles INTO wa_flight.
      MOVE-CORRESPONDING wa_flight TO wa_col_flight.
      wa_col_flight-color = col_background.
      INSERT wa_col_flight INTO TABLE it_col_doubles.
    ENDLOOP.
    CALL FUNCTION 'BC401_ALV_LIST_OUTPUT'
      EXPORTING
        it_list1      = it_col_flights
        it_list2      = it_col_doubles
      EXCEPTIONS
        control_error = 1
        OTHERS        = 2.
    IF sy-subrc <> 0.
      MESSAGE a702(bc401).
    ENDIF.
  ELSE.
    LOOP AT it_flights INTO wa_flight.
      IF wa_flight-fldate < sy-datum.
        FORMAT COLOR = col_negative.
      ELSEIF wa_flight-fldate < pa_date.
        FORMAT COLOR = col_total.
      ELSE.
        FORMAT COLOR = col_positive.
      ENDIF.
      WRITE: /
        wa_flight-carrid,
        wa_flight-connid,
        wa_flight-fldate DD/MM/YYYY,
        wa_flight-price CURRENCY wa_flight-currency,
        wa_flight-currency,
        wa_flight-planetype,
        wa_flight-seatsmax,
        wa_flight-seatsocc.
    ENDLOOP.
    FORMAT RESET.
    SKIP.
    WRITE: / 'duplicate data records:'(dob) COLOR COL_HEADING.
    LOOP AT it_doubles INTO wa_flight.
      WRITE: /
        wa_flight-carrid,
        wa_flight-connid,
        wa_flight-fldate DD/MM/YYYY,
        wa_flight-price CURRENCY wa_flight-currency,
        wa_flight-currency,
        wa_flight-planetype,
        wa_flight-seatsmax,
        wa_flight-seatsocc.
    ENDLOOP.

